
#1: Beating Up at Service Discovery

No one cares if one instance is down:

up{job="foo"} == 0

Are at least 3 instances up?

sum(up{job="foo"}) < 3

A ratio of healthy instances to total:

avg(up{job="foo"}) < 0.8

When service discovery takes your lunch:

sum(up{job="foo"}) < 3 OR absent(up{job="foo"})

#2: Who’s Bouncing Your Apps?

No one cares if an instance restarted 5 minutes ago:

time() - process_start_time_seconds < 300

Instances in a restart loop:

changes(process_start_time_seconds[1h]) > 3

All of your instances are restarting:

avg WITHOUT (instance) (
changes(process_start_time_seconds[1h]) > bool 3

) > 0.5

2



#3: Using build_info Like a Pro

Every application should export a build_info metric:
prometheus_build_info{goversion="go1.12.5",revision="<hash>",version="2.10.0"} 1

Shove the version label into other metrics:

go_memstats_alloc_bytes * on (instance)
group_left(version) prometheus_build_info

Median memory usage across all instances:

quantile(0.5,
go_memstats_alloc_bytes * on (instance)
group_left(version) prometheus_build_info

) WITHOUT (instance)

#4: Cheater’s Guide to Histograms and SLOs

Make the SLO a parameter to the application and export it!
foobar_slo_seconds 10

Use these 3 Counters:
foobar_pipeline_events_total{handler="foo",client="iOS"} 42
foobar_pipeline_events_errors_total{handler="foo",client="iOS"} 3
foobar_pipeline_events_over_slo_total{handler="foo",client="iOS"} 1

SLO alert for 90% of events processed under 10 seconds for last 24 hours:

1 - rate(foobar_pipeline_events_over_slo_total[1d]) /
rate(foobar_pipeline_events_total[1d])

< 0.9

#5: Estimating Storage Space

Efficiency of the Gorilla algorithm:

rate(prometheus_tsdb_compaction_chunk_size_bytes_sum[1d])
/
rate(prometheus_tsdb_compaction_chunk_samples_sum[1d])

Median:

quantile(0.5,
rate(prometheus_tsdb_compaction_chunk_size_bytes_sum[1d])
/
rate(prometheus_tsdb_compaction_chunk_samples_sum[1d])

) WITHOUT (instance)

Thank You!

